Constrained Minima

```
> restart:
> with(LinearAlgebra):
  with(plots):
> f  := x+y+x*z ;
  h1 := z ;
  h2 := x^2+y^2-z-1 ;
```

$$f := x + y + x\,z$$
$$h1 := z$$
$$h2 := x^2 + y^2 - z - 1 \quad (1)$$

Step 1: Build the Lagrangian

```
> L := f-lambda1*h1-lambda2*h2 ;
```

$$L := x + y + x\,z - \lambda 1\, z - \lambda 2\left(x^2 + y^2 - z - 1\right) \quad (2)$$

Step 2: Build non linear system

```
> EQ1 := diff( L, x ) ;
  EQ2 := diff( L, y ) ;
  EQ3 := diff( L, z ) ;
  EQ4 := diff( L, lambda1 ) ;
  EQ5 := diff( L, lambda2 ) ;
```

$$EQ1 := 1 + z - 2\,\lambda 2\, x$$
$$EQ2 := 1 - 2\,\lambda 2\, y$$
$$EQ3 := x - \lambda 1 + \lambda 2$$
$$EQ4 := -z$$
$$EQ5 := -x^2 - y^2 + z + 1 \quad (3)$$

Step 3: Solve the non linear system

```
> solve( {EQ||(1..5)}, {x,y,z,lambda1,lambda2} ) :
  SOL := allvalues(%) ;
```

$$SOL := \left\{\lambda 1 = \sqrt{2}, \lambda 2 = \frac{1}{2}\sqrt{2}, x = \frac{1}{2}\sqrt{2}, y = \frac{1}{2}\sqrt{2}, z = 0\right\}, \left\{\lambda 1 = -\sqrt{2}, \lambda 2 = -\frac{1}{2}\sqrt{2}, x = -\frac{1}{2}\sqrt{2}, y = -\frac{1}{2}\sqrt{2}, z = 0\right\} \quad (4)$$

Step 4: compute the kernel of the constraints

```
> H := <<diff(h1,x),diff(h2,x)>|
        <diff(h1,y),diff(h2,y)>|
        <diff(h1,z),diff(h2,z)>>;
```

$$H := \begin{bmatrix} 0 & 0 & 1 \\ 2\,x & 2\,y & -1 \end{bmatrix} \quad (5)$$

```
> Hess := <<diff(L,x,x),diff(L,y,x),diff(L,z,x)>|
           <diff(L,x,y),diff(L,y,y),diff(L,z,y)>|
           <diff(L,x,z),diff(L,y,z),diff(L,z,z)>>;
```

$$Hess := \begin{bmatrix} -2\,\lambda 2 & 0 & 1 \\ 0 & -2\,\lambda 2 & 0 \\ 1 & 0 & 0 \end{bmatrix} \quad (6)$$

First point

```
> H1     := subs( SOL[1], H) ;
  Hess1 := subs( SOL[1], Hess) ;
```

$$H1 := \begin{bmatrix} 0 & 0 & 1 \\ \sqrt{2} & \sqrt{2} & -1 \end{bmatrix}$$

$$Hess1 := \begin{bmatrix} -\sqrt{2} & 0 & 1 \\ 0 & -\sqrt{2} & 0 \\ 1 & 0 & 0 \end{bmatrix} \quad (7)$$

```
> Z1 := alpha*op(NullSpace(H1)) ;
```

$$Z1 := \begin{bmatrix} -\alpha \\ \alpha \\ 0 \end{bmatrix} \quad (8)$$

Is a max point

```
> Transpose(Z1).Hess1.Z1 ;
```

$$-2\,\alpha^2\,\sqrt{2} \quad (9)$$

Second point

```
> H2     := subs( SOL[2], H) ;
  Hess2 := subs( SOL[2], Hess) ;
```

$$H2 := \begin{bmatrix} 0 & 0 & 1 \\ -\sqrt{2} & -\sqrt{2} & -1 \end{bmatrix}$$

$$Hess2 := \begin{bmatrix} \sqrt{2} & 0 & 1 \\ 0 & \sqrt{2} & 0 \\ 1 & 0 & 0 \end{bmatrix} \quad (10)$$

```
> Z2 := alpha*op(NullSpace(H2)) ;
```

$$Z2 := \begin{bmatrix} -\alpha \\ \alpha \\ 0 \end{bmatrix} \quad (11)$$

Is a min point

```
> Transpose(Z2).Hess2.Z2 ;
```

$$2\,\alpha^2\,\sqrt{2} \quad (12)$$

```
> P1 := plot3d( subs(z=0,f),x=-2..2,y=-2..2) ;
```

$$P1 := PLOT3D(\ldots) \tag{13}$$

```
> P2 := spacecurve( [ sin(t),cos(t),subs(x=cos(t),y=sin(t),z=0,f)],
  t=0..2*Pi,thickness=3,color=black);
```

$$P2 := PLOT3D(\ldots) \tag{14}$$

```
> P3 := pointplot3d( {subs(SOL[1],[x,y,f]),subs(SOL[2],[x,y,f])},
  symbolsize=20, color=red ):
```

```
> display(P1,P2,P3) ;
```




```
> plot( subs(x=cos(t),y=sin(t),z=0,f), t=0..2*Pi ) ;
```


1
0.5
0
−0.5
−1
π/4
π/2
3 π/4
π
5 π/4
3 π/2
7 π/4
2 π
t